

Type families and data kinds

Advanced functional programming - Lecture 8

Wouter Swierstra and Alejandro Serrano



Last lecture

In the previous lecture, we saw some examples of programming with GADTs.

- ▶ This allowed us to define rich types, enforcing all kinds of properties.
- ▶ But we also ran into some limitations.



Today

- ▶ A bit more on the `Equal` data type
- ▶ Kinds beyond `*`
- ▶ Programming with types



Equality type

We introduce `Equal` as a proof that two types are equal:

```
data Equal :: * -> * -> * where  
  Refl :: Equal a a
```

We could even 'prove' some properties of the relation:

```
refl  :: Equal a a  
sym   :: Equal a b -> Equal b a  
trans :: Equal a b -> Equal b c -> Equal a c
```



Expressive power of equality

Equality proofs and phantom types are enough to 'implement' GADTs:

```
data Expr a =  
  LitI   (Equal a Int)   Int  
| LitB   (Equal a Bool)  Bool  
| IsZero (Equal a Bool)  (Equal b Int)  
| Plus   (Equal a Int)   (Expr Int) (Expr Int)  
| If     (Expr Bool)     (Expr a)  (Expr a)
```



Safe vs unsafe coercions

Using our equality function we can safely coerce between types:

```
coerce :: Equal a b -> a -> b  
coerce Refl x = x
```

Question

Why does this type check?



Safe vs unsafe coercions

Using our equality function we can safely coerce between types:

```
coerce :: Equal a b -> a -> b
coerce Refl x = x
```

Question

Why does this type check?

Question

What about this definition:

```
coerce :: Equal a b -> a -> b
coerce p x = x
```



Aside: irrefutable patterns

Haskell also allows irrefutable patterns:

`lazyHead ~(x:xs) = x`

This does not force the list to weak head normal form.



Aside: irrefutable patterns

In tandem with GADTs this is particularly dangerous:

```
coerceL :: Equal a b -> a -> b  
coerceL ~Refl x = x
```

Question

How could this cause well-typed program to crash with a type error?



Aside: irrefutable patterns

In tandem with GADTs this is particularly dangerous:

```
coerceL :: Equal a b -> a -> b  
coerceL ~Refl x = x
```

Question

How could this cause well-typed program to crash with a type error?

```
foo :: Bool -> Int  
foo b = coerceL undefined b
```

Apparently unrelated language features may interact in unexpected ways!



Outlook generic programming

We can even use GADTs to *reflect* types themselves as data:

```
data Type :: * -> * where
  INT  :: Type Int
  BOOL :: Type Bool
  LIST :: Type a -> Type [a]
  PAIR :: Type a -> Type b -> Type (a,b)
```



Safe dynamically typed values

We can define dynamically typed values by packing up a type representation with a value:

```
data Dynamic :: * where  
  Dyn :: Type a -> a -> Dynamic
```



Safe dynamically typed values

We can define dynamically typed values by packing up a type representation with a value:

```
data Dynamic :: * where
  Dyn :: Type a -> a -> Dynamic
```

To unwrap these values safely, we check whether the types line up as expected:

```
coerce :: Type a -> Dynamic -> Maybe a
coerce t (Dyn t' x) =
  case eqType t t'
    of Just Refl -> Just x
       _         -> Nothing
```



Generic programming

We can also define new functions *by induction on the type structure*:

$f :: \text{Type } a \rightarrow \dots a \dots$

In this way, we can define our own versions of functions such as `show`, `read`, `equality`, etc.



System FC

Haskell's core language, System FC, is a typed lambda calculus, extended with data types and pattern matching.

One of its more distinct features is *coercions* and *casts*.

- ▶ Coercions play the same role as our `Equal` data type;
- ▶ If two types are coercible, one can be cast to the other:

```
isZero :: (a ~ Int) => a -> Bool
```

There is quite a lot of work necessary to guarantee that this does not accidentally make the type system unsound!

Pattern matching on GADTs introduces such coercions in the individual branches.



Problems with GADTs

```
toVec :: [a] -> Vec a ???
```

To define this function, we needed to reify natural numbers on the type level – defining a *singleton type* *SNat*.

```
vappend :: Vec a n -> Vec a m -> Vec a ???
```

To define this function, we needed to construct an explicit relation describing how to add two types, *n* and *m*.



Passing explicit Sums

In the last lecture, we saw how to pass explicit an explicit argument, explaining how to add two 'type-level' natural numbers:

```
data Sum :: * -> * -> * -> * where
  SumZero :: Sum Zero n n
  SumSucc  :: Sum n m s -> Sum (Succ n) m (Succ s)
```

But constructing this evidence by hand is tedious...



Multi-parameter type classes

We can automate this through a *multi-parameter type class*

```
class Summable a b c | a b -> c where  
  makeSum :: Sum a b c
```

```
instance Summable Zero n n where  
  makeSum = SumZero
```

```
instance Summable n m s =>  
  Summable (Succ n) m (Succ s) where  
  makeSum = SumSucc makeSum
```

```
append :: Sum n m s  
  => Vec a n -> Vec a m -> Vec a s  
append = vappend makeSum
```



Multi-parameter type classes

Type classes define *relations* between types:

- ▶ `Eq` defines a subset of all types that support equality;
- ▶ `MonadState` (from `mt1`) defines a subset of pairs of types `s` and `m`, where `m` supports read/write operations on a state of type `s`.

The `Summable` type class is special case of such relations – it is really defining a *function* between types.



Multi-parameter type classes

For some time, multi-parameter type classes with functional dependencies were the *only* way in Haskell to define such type-level computations.

But there has been a flurry of research in the last decade exploring alternative language extensions.

... the interaction of functional dependencies with other type-level features such as existentials and GADTs is not well understood and possibly problematic.

Kiselyov, Peyton Jones, Shan in *Fun with type families*



Associated types and type families

Type classes let you capture an *interface* – such as monads (supporting return and bind), or monoids (supporting zero and addition).

These interfaces can describe *functions*.

But what if we would like them to describe *types*.



Associated types

Associated types let you declare a type in a class declaration:

```
class Collects c where
  type Elem c -- Associated type synonym
  empty  :: c
  insert :: Elem c -> c -> c
  toList :: c -> [Elem c]
```

Any instance of the `Collects` class must choose a type of elements, together with definitions for the functions.



Associated types – examples

```
instance Eq e => Collects [e] where
  type Elem [e] = e
  empty = []
  ...
```

```
instance Collects BitSet where
  type Elem BitSet = Char
  ...
```



Addition through association

We can use such associated types to replace the functional dependencies we saw previously:

```
class Summable n m where
  type TheSum n m
  makeSum :: Sum n m (TheSum n m)
```

```
instance Summable Zero n where
  type TheSum Zero m = m
  ...
```

```
instance Summable n m => Summable (Succ n) m where
  type TheSum (Succ n) m = Succ (TheSum n m)
  ...
```



Associated types or multiparameter type?

Both approaches are similar in expressive power.

Multiparameter type classes are no longer fashionable

- ▶ They can make type class resolution unpredictable.

Associated types have gained traction in other languages

- ▶ Like Rust or Apple's Swift.



Type families

Associated types always require a class definition – even if we're only interested in the types.

Type families build upon the technology that associated types provide, enabling you to write:

```
type family Sum n m
type instance Sum Zero      n = n
type instance Sum (Succ n) m = Succ (Sum n m)
```

This looks much more like regular programming...



Closed type families

If we piggyback on the associated type machinery, however, all our type families are *open* – we can add bogus definitions:

```
type instance Sum n Zero = Zero
```



Closed type families

If we piggyback on the associated type machinery, however, all our type families are *open* – we can add bogus definitions:

```
type instance Sum n Zero = Zero
```

The more modern *closed type families* allow you to define a function between types using pattern matching:

```
type family Sum n m where  
  Sum Zero      n = n  
  Sum (Succ n) m = Succ (Sum n m)
```

GHC will try to match a given type against the patterns one by one, taking the first branch that matches successfully.



Apartness for closed type families

GHC only moves to the next branch if it know that the previous one may never match.

- ▶ The types are *apart* from the pattern.

```
> :kind! Sum n (Succ m)  
-- does not reduce further!
```



The need for more kinds

Furthermore, all our ‘type level’ code is essentially untyped.

```
type instance Sum Bool Int = Char
```

We want “type-level types”, as we have in the term-level.



So far we have seen that two different forms of kinds:

- ▶ all types have kind $*$
- ▶ given two kinds k_1 and k_2 , we can form the kind $k_1 \rightarrow k_2$ – corresponding to the type constructor taking something of kind k_1 to produce a type of kind k_2 .



So far we have seen that two different forms of kinds:

- ▶ all types have kind $*$
- ▶ given two kinds k_1 and k_2 , we can form the kind $k_1 \rightarrow k_2$ – corresponding to the type constructor taking something of kind k_1 to produce a type of kind k_2 .

As soon as we do richer programming with types, however, we would like stronger guarantees about the safety of the *type level* computations that we write.



Promotion

```
data Zero
```

```
data Succ n
```

```
data Nat = Zero | Succ Nat
```

How can we ensure all numbers in our types to be built from Zero and Succ?



Promotion

Using the DataKinds language extension we can introduce new kinds and automatically *promote* data constructors into their type-level variants:

```
{-# LANGUAGE DataKinds #-}
```

```
data Nat = Zero | Succ Nat
```

This declaration introduces:

- ▶ a new kind `Nat`
- ▶ a type `'Zero :: Nat`
- ▶ a type `'Succ :: Nat -> Nat`

If you want to promote a GADT, you need `TypeInType`.



Example: booleans

```
-- the usual definition of booleans
data Bool = True | False

-- Not function on values
not :: Bool -> Bool
not True = False
not False = True

-- Not function on types
type family Not (a :: Bool) :: Bool where
    Not True = False
    Not False = True
```



Type-level literals

GHC takes the idea of programming with types quite far.
It has added support and syntax for:

- ▶ type-level strings;
- ▶ type-level lists;
- ▶ type-level integers;



List membership

Member `x xs` should be true if `x` is a member of the type-level list `xs`.

First attempt: type class

```
class    Member x xs
instance Member x (x ': xs)
instance Member x xs => Member x (y ': ys)
```

Why is this code rejected?



List membership

Member x xs should be true if x is a member of the type-level list xs.

First attempt: type class

```
class    Member x xs
instance Member x (x ': xs)
instance Member x xs => Member x (y ': ys)
```

Why is this code rejected?

```
instance {-# OVERLAPS #-}
    Member x (x ': xs)
instance {-# OVERLAPPABLE #-}
    Member x xs => Member x (y ': ys)
```



The case against overlapping instances

1. Overlapping instances make type resolution brittle.
 - ▶ Which instance is selected depends on how much we know about a type.
2. Overlapping instances are not modular.
 - ▶ Adding a new instance that overlaps may render previous resolution wrong.



List membership with a closed type family

```
type family Member' x xs where
  Member' x '[]          = 'False
  Member' x (x ': xs)    = 'True
  Member' x (y ': xs)    = Member' x xs
```

```
type Member x xs = Member' x xs ~ 'True
```

- ▶ The second branch is *non-linear*, x is repeated.
 - ▶ Not allowed in term-level pattern matching.
 - ▶ Fine with closed type families.
- ▶ Member defines a synonym for a constraint.
 - ▶ In GHC, constraints are just types of the special Constraint kind.



Polymorphic kinds

Which is the kind of `Member` '?

`Member` ' $:: *$ $\rightarrow [*]$ $\rightarrow \text{Bool}$

`Member` ' $:: \text{Nat}$ $\rightarrow [\text{Nat}] \rightarrow \text{Bool}$



Polymorphic kinds

Which is the kind of `Member'`?

`Member' :: * -> [*] -> Bool`

`Member' :: Nat -> [Nat] -> Bool`

Kinds may be *polymorphic*, as types are!

`Member' :: k -> [k] -> Bool`



Type-level application

```
data Apply f a = MkApply (f a)
```

Question

What is the kind of Apply?

```
> :set -XPolyKinds
```

```
> :info Apply
```

```
type role Apply representational nominal
```

```
data Apply (f :: k -> *) (a :: k) = MkApply (f a)
```



Proxy data types

Consider the read function:

```
read :: Read a => String -> a
```

How do we fix the type we want to get?



Proxy data types

Consider the read function:

```
read :: Read a => String -> a
```

How do we fix the type we want to get?

First solution, annotate it

```
read "123" :: Int
```

In many cases, the type can be inferred instead.



Proxy data types

Second solution, fix it via other parameter

```
-- Wrap it with an additional argument
read' :: Read a => a -> String -> a
read' _ = read
-- And use it like this
read' (undefined :: Int) "123"
```

- ▶ The first argument is never touched.
- ▶ Gives valuable information to the compiler.

These are called *proxy* arguments.



Proxy data types

```
-- Polymorphic kind!  
data Proxy (a :: k) = Proxy  
-- The proxy is also ignored  
read' :: Read a => Proxy a -> String -> a  
read' _ = read  
-- To use it you create a Proxy value  
read (Proxy :: Proxy Int) "123"
```

Question

What is the benefit of using Proxy?



(No) proxy data types

Third solution, explicit type application

```
read @Int "123"
```

Only from GHC 8 with TypeApplications enabled.



Outlook: writing webserver with Servant

Servant is a library for describing web APIs. From such a description, it will generate documentation, a simple webserver, a JavaScript client, etc.

- ▶ Instead of describing the APIs using Haskell values – you describe the API as a (complex) Haskell type.
- ▶ And then *generate* any desired functionality from it.

```
type HackageAPI =  
    "users"  :> Get '[JSON] [UserSummary]  
  :<|> "user" :> Capture "username" Username  
        :> Get '[JSON] UserDetailed  
  :<|> "packages" :> Get '[JSON] [Package]
```



Recap: GADTs

GADTs give you more power to define interesting types.

- ▶ We can decorate our types with *more specific information*.

But we still cannot do any interesting *computation*.

- ▶ We need to use type classes or type families.
- ▶ We end up mirroring expression-level concepts on the type level (e.g. natural numbers).

The ‘value language’ and ‘type language’ live in very different worlds...

